

### Θέμα 3: Αναζήτηση υπόπτου

Η άτυχη Σουζάνα βρέθηκε από την αστυνομία δολοφονημένη, με τσακισμένο το κεφάλι από αμβλύ όργανο. Τα πρόσωπα της ιστορίας είναι ο Άλαν, 35 ετών, χασάπης και «κλεφτρόνι», ο Τζον, 25 ετών, ποδοσφαιριστής, που συνδέεται συναισθηματικά τόσο με τη δολοφονημένη Σουζάνα όσο και με τη Μπάρμπαρα, κομμώτρια, 22 ετών και παντρεμένη με τον πενηντάρη και κουτσό μαραγκό Μπερτ. Σε σχετική έρευνα ανακαλύφθηκε στο σπίτι του Τζον ένα περίστροφο. Η αστυνομία θεωρεί ως πιθανά κίνητρα τη ληστεία και τη ζήλεια.

#### Θέμα 3α. Αναπαράσταση γεγονότων του προβλήματος

Να αναπαραστήσετε τα γεγονότα του προβλήματος χρησιμοποιώντας την δηλωτική γλώσσα προγραμματισμού SWI-Prolog (<http://www.swi-prolog.org/>), λαμβάνοντας υπόψη τα παρακάτω:

- Χρησιμοποιήστε τα ακόλουθα κατηγορήματα:  
`person/4, had_affair/2, motive/1, owns/2.`
- Τα τέσσερα ορίσματα του `person/4` αναφέρονται στο όνομα, ηλικία, φύλο και ιδιότητα του προσώπου που περιγράφει, π.χ.  
`person(john,25,m,football_player).`
- Τα δύο ορίσματα του `had_affair/2` δηλώνουν τα πρόσωπα που έχουν/είχαν δεσμό.
- Δίνεται το όργανο με το οποίο σκοτώθηκε η Σουζάνα, δηλ.  
`killed_with(susan,club).`
- Ο Μπερτ υποκαθιστά το χαμένο του πόδι με ένα ξύλινο.
- Δίνονται οι εξής υποθέσεις κοινής λογικής  
`operates_identically(wooden_leg,club).`  
`operates_identically(bar,club).`  
`operates_identically(pair_of_scissors,knife).`  
`operates_identically(football_boot,club).`

#### Ενδεικτική επίλυση

```
%-----  
% Τύποι ισχυρισμών και υποθέσεων από  
% την έρευνα της αστυνομίας.  
%-----  
person(john,25,m,football_player). /*m:male*/  
person(allan,35,m,butcher).  
person(barbara,22,f,hairstylist). /*f:female*/  
person(bert,50,m,carpenter).  
person(allan,35,m,pickpocket).  
  
had_affair(barbara,john).  
had_affair(barbara,bert).  
had_affair(susan,john).  
  
killed_with(susan,club).  
  
motive(money).  
motive(jealousy).  
  
owns(bert,wooden_leg).
```

```
owns(john,pistol).
```

```
%-----  
% Υποθέσεις κοινής λογικής  
%-----  
operates_identically(wooden_leg,club).  
operates_identically(bar,club).  
operates_identically(pair_of_scissors,knife).  
operates_identically(football_boot,club).
```

### Θέμα 3β. Αναπαράσταση κανόνων κοινής λογικής

Περιγράψτε με προτάσεις Κατηγορηματικής Λογικής και κώδικα Prolog, χρησιμοποιώντας το κατηγορήμα `owns_probably/2` καθώς και κατηγορήματα από το Θέμα 2α, την παρακάτω κοινά αποδεκτή γνώση:

- Ένας ποδοσφαιριστής κατέχει πιθανώς μπότα ποδοσφαίρου (το σενάριο αναφέρεται στο αμερικάνικο ποδόσφαιρο)
- Οποιοσδήποτε μπορεί να κατέχει ένα ψαλίδι.
- Οποιοσδήποτε μπορεί να κατέχει ένα μαχαίρι.
- Όποιος κατέχει ένα αντικείμενο τότε πιθανώς κατέχει αυτό το αντικείμενο.

#### Ενδεικτική επίλυση

*Προτάσεις Κατηγορηματικής Λογικής*

$$\forall x, y, z (\text{owns\_probably}(x, \text{football\_boot}) \Leftarrow \text{person}(x, y, z, \text{football\_player}))$$
$$\forall x, y, z, w (\text{owns\_probably}(x, \text{pair\_of\_scissors}) \Leftarrow \text{person}(x, y, z, w))$$
$$\forall x, y, z, w (\text{owns\_probably}(x, \text{knife}) \Leftarrow \text{person}(x, y, z, w))$$
$$\forall x, y (\text{owns\_probably}(x, y) \Leftarrow \text{owns}(x, y))$$

#### Κώδικας SWI-Prolog

```
owns_probably(X, football_boot):- person(X,_,_, football_player).  
owns_probably(X, pair_of_scissors):- person(X,_,_,_).  
owns_probably(X, knife):- person(X,_,_,_).  
owns_probably(X, Object):- owns(X, Object).
```

### Θέμα 3γ. Αναπαράσταση κανόνων επίλυσης του προβλήματος

Δημιουργείστε κανόνες επίλυσης του προβλήματος θεωρώντας ότι υπάρχουν δύο τύποι υπόπτων: α) ύποπτοι με βάση τη δυνατότητα να διαπράξουν τον φόνο, και β) ύποπτοι με βάση το κίνητρό τους. Οι υποθέσεις για τα κίνητρα του φόνου είναι οι εξής:

Κάποιος X είναι ύποπτος με βάση τη δυνατότητα να διαπράξει τον φόνο αν το αντικείμενο με το οποίο σκοτώθηκε η Σουζάνα έχει την ίδια λειτουργία (ίδια αποτελέσματα) με κάποιο άλλο αντικείμενο Y και ο X πιθανώς κατέχει το αντικείμενο Y.

Κάποιος X είναι ύποπτος με βάση το κίνητρό του αν

1. το κίνητρο είναι η ζήλεια, ο X είναι άνδρας και είχε δεσμό με τη Σουζάνα
2. το κίνητρο είναι η ζήλεια, ο X είναι γυναίκα και είχε δεσμό κάποιον ο οποίος είχε ταυτόχρονα δεσμό με τη Σουζάνα

3. το κίνητρο είναι τα χρήματα και ο X είναι «κλεφτρόνι».

Ο πιο πιθανός ύποπτος είναι αυτός που είναι ταυτόχρονα ύποπτος με βάση τη δυνατότητα και το κίνητρο να διαπράξει τον φόνο.

**Συμβουλή:** Χρησιμοποιήστε τα ακόλουθα κατηγορήματα:

suspect\_by\_capability/1, suspect\_by\_motive/1, mostly\_suspected/1.

### Ενδεικτική επίλυση

Συνεχίζοντας το πρόγραμμα που δίνεται στην ενδεικτική επίλυση του 2α, έχουμε:

```
%-----
% Υποθέσεις για τα κίνητρα φόνου.
% Δύο τύποι υπόπτων:
% α. ύποπτοι με βάση τη δυνατότητα να
%   διαπράξουν τον φόνο
% β. ύποπτοι με βάση το κίνητρό τους
%-----

suspect_by_capability(X):-
    killed_with(susan,Weapon),
    operates_identically(Object,Weapon),
    owns_probably(X,Object).

suspect_by_motive(X):-
    motive(jealousy),
    person(X,_,m,_),
    had_affair(susan,X).

suspect_by_motive(X):-
    motive(jealousy),
    person(X,_,f,_),
    had_affair(X,Man),
    had_affair(susan,Man).

suspect_by_motive(X):-
    motive(money),
    person(X,_,_,pickpocket).

%-----
% Μη πιθανοθεωρητική υπόθεση κοινής λογικής
%-----

mostly_suspected(X):-
    suspect_by_capability(X),
    suspect_by_motive(X).
```

### Θέμα 3δ. Εκτέλεση του προγράμματος

Εκτελέστε το παραπάνω πρόγραμμα κάνοντας κατάλληλα ερωτήματα (queries) ώστε να βρίσκει τον πιο πιθανό ύποπτο.

### Ενδεικτική επίλυση

```
1 ?- mostly_suspected(X).
```

```
X = john
```